


Examples of use of IDL SPEDAS code for EFW

Spin-fit electric field field and related data
The spin-fit electric field and related data are saved in the L3 data. Below are some example IDL codes for loading the commonly used data including the spin-fit E field, the EFW density calibrated according to the upper-hybrid line, and the spacecraft potential. We note that in order to ensure backward compatibility, the L2 spin-fit data is kept. However, because the L2 data only contains a subset of the variables in L3, we strongly encourage users to use the L3 data instead of L2.

time_range = ['2013-01-01','2013-01-03']
probe = 'a'

; Load the spin-fit data.
rbsp_efw_read_l3, time_range, probe=probe

prefix = 'rbsp'+probe+'_efw_'
vars = prefix+[$

 ; The spin-fit E field with E_spinaxis = 0.
'efield_in_corotation_frame_spin-fit_mgse', $

; The spin-fit E field with E_spinaxis calculated from E dot B = 0, 
; when B is away from the spin plane by >15 deg.
'efield_in_corotation_frame_spin-fit_edotb_mgse', $

; The spacecraft potential.
'spacecraft_potential', $

; The EFW density calibrated according to the upper-hybrid line.
'density', $

; The ephemeris data.
'position_gse', 'velocity_gse', 'mlt', 'mlat', 'lshell', 'orbit_num' ]

; Plot the variables.
tplot, vars, trange=time_range


Survey E field and related data
The survey-mode electric fields are saved in the L2 esvy_despun and vsvy data.

; Set the time and probe for loading data.
time_range = ['2013-01-01','2013-01-03']
probe = 'b'

; Load the L2 data.
rbsp_efw_read_l2, time_range, probe=probe, datatype='esvy_despun'
rbsp_efw_read_l2, time_range, probe=probe, datatype='vsvy-hires'

prefix = 'rbsp'+probe+'_efw_'
vars = prefix+[$

; The E field in mGSE.
'efield_mgse', $

; The single-ended probe potential.
'vsvy', $

; The flags.
'efw_qual' ]

; Plot the variables.
tplot, vars, trange=time_range


Burst data
Here are some example IDL codes to load the burst electric and magnetic field, with preliminary calibrations.

; Set the time and probe for loading data.
time_range = ['2013-06-10/05:57:20','2013-06-10/05:59:40']
probe = 'b'

; Load the burst data.
rbsp_efw_read_burst_efield, time_range, probe=probe, datatype='vb1', coord='mgse'
rbsp_efw_read_burst_bfield, time_range, probe=probe, datatype='mscb1', coord='mgse'

prefix = 'rbsp'+probe+'_efw_'
vars = prefix+[$

; The E and B fields in mgse.
['eb1','mscb1']+'_mgse' ]

; Plot the variables.
tplot, vars, trange=time_range


Supporting data
Here are some example IDL codes to load the bias current, to rotate the electric field from the UVW or mGSE to the GSE coordinate.

; Set the time and probe for loading data.
;time_range = ['2018-10-08/02:00','2018-10-08/03:30']
time_range = ['2013-03-17/09:55','2013-03-17/10:05']
probe = 'b'

; Other settings.
uvw = ['u','v','w']
xyz = ['x','y','z']
rgb = [6,4,2]
tplot_options, 'labflag', -1
prefix = 'rbsp'+probe+'_efw_'

; Rotate E field from mGSE to GSE.
rbsp_efw_read_l3, time_range, probe=probe
e_mgse_var = prefix+'efield_in_corotation_frame_spin-fit_mgse'
get_data, e_mgse_var, times, e_mgse
store_data, e_mgse_var, limits={ytitle:'[mV/m]', labels:'mGSE E'+xyz, colors:rgb}
; Require to set probe when the rotation involves mgse and uvw.
e_gse = cotran(e_mgse, times, 'mgse2gse', probe=probe)
store_data, prefix+'e_gse', times, e_gse, limits={ytitle:'[mV/m]', labels:'GSE E'+xyz, colors:rgb}

; Rotate E field from UVW to GSE.
rbsp_efw_read_l2, time_range, probe=probe, datatype='e-hires-uvw'
e_uvw_var = prefix+'e_hires_uvw'
get_data, e_uvw_var, times, e_uvw
e_uvw[*,2] = 0
store_data, e_uvw_var, limits={labels: 'E'+uvw, colors:rgb}
; Require to set probe when the rotation involves mgse and uvw.
e_gse = cotran(e_uvw, times, 'uvw2gse', probe=probe)
store_data, prefix+'e_gse2', times, e_gse, limits={ytitle:'[mV/m]', labels:'GSE E'+xyz, colors:rgb}

; Plot data.
options, prefix+'bias_current', 'labels', string(findgen(6)+1,format='(I0)')

vars = prefix+[$
; The bias current.
'bias_current', $
; E fields in various coords.
'efield_in_corotation_frame_spin-fit_mgse', $
'e_hires_uvw', 'e_gse', 'e_gse2' ]

